NAG Fortran Library Routine Document

G01JCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G01JCF returns the lower tail probability of a distribution of a positive linear combination of χ^2 random variables.

2 Specification

```
SUBROUTINE G01JCF(A, MULT, RLAMDA, N, C, P, PDF, TOL, MAXIT, WRK, IFAIL)
INTEGER           MULT(N), N, MAXIT, IFAIL
real              A(N), RLAMDA(N), C, P, PDF, TOL, WRK(N+2*MAXIT)
```

3 Description

For a linear combination of non-central χ^2 random variables with integer degrees of freedom the lower tail probability is

$$P\left(\sum_{j=1}^{n} a_j \chi^2(m_j, \lambda_j) \leq c\right), \tag{1}$$

where a_j and c are positive constants and where $\chi^2(m_j, \lambda_j)$ represents an independent χ^2 random variable with m_j degrees of freedom and non-centrality parameter λ_j. The linear combination may arise from considering a quadratic form in Normal variables.

Ruben's method as described in Farebrother (1984) is used. Ruben has shown that (1) may be expanded as an infinite series of the form

$$\sum_{k=0}^{\infty} d_k F(m + 2k, c/\beta), \tag{2}$$

where $F(m + 2k, c/\beta) = \mathrm{P}\big(\chi^2(m + 2k) < c/\beta\big)$, i.e., the probability that a central χ^2 is less than c/β.

The value of β is set at

$$\beta = \beta_B = \frac{2}{(1/a_{\min} + 1/a_{\max})}$$

unless $\beta_B > 1.8a_{\min}$, in which case

$$\beta = \beta_A = a_{\min}$$

is used, where $a_{\min} = \min\{a_j\}$ and $a_{\max} = \max\{a_j\}$, for $j = 1, 2, \ldots, n$.

4 References

Farebrother R W (1984) The distribution of a positive linear combination of χ^2 random variables *Appl. Statist.* **33 (3)**

5 Parameters

1: A(N) – ***real*** array *Input*

On entry: the weights, $a_1, a_2, \ldots, a_n$.

Constraint: $\mathrm{A}(i) > 0.0$, for $i = 1, 2, \ldots, n$.

2: MULT(N) – INTEGER array *Input*

On entry: the degrees of freedom, $m_1, m_2, \ldots, m_n$.

Constraint: $\mathrm{MULT}(i) \geq 1$, for $i = 1, 2, \ldots, n$.

3: RLAMDA(N) – ***real*** array *Input*

On entry: the non-centrality parameters, $\lambda_1, \lambda_2, \ldots, \lambda_n$.

Constraint: $\mathrm{RLAMDA}(i) \geq 0.0$, for $i = 1, 2, \ldots, n$.

4: N – INTEGER *Input*

On entry: the number of χ^2 random variables in the combination, n, i.e., the number of terms in equation (1).

Constraint: $\mathrm{N} \geq 1$.

5: C – ***real*** *Input*

On entry: the point for which the lower tail probability is to be evaluated, c.

Constraint: $\mathrm{C} \geq 0.0$.

6: P – ***real*** *Output*

On exit: the lower tail probability associated with the linear combination of n χ^2 random variables with m_j degrees of freedom, and non-centrality parameters λ_j, for $j = 1, 2, \ldots, n$.

7: PDF – ***real*** *Output*

On exit: the value of the probability density function of the linear combination of χ^2 variables.

8: TOL – ***real*** *Input*

On entry: the relative accuracy required by the user in the results. If G01JCF is entered with TOL greater than or equal to 1.0 or less than 10 times the ***machine precision*** (see X02AJF), then the value of 10 times ***machine precision*** is used instead.

9: MAXIT – INTEGER *Input*

On entry: the maximum number of terms that should be used during the summation.

Suggested value: 500.

Constraint: $\mathrm{MAXIT} \geq 1$.

10: WRK(N+2*MAXIT) – ***real*** array *Workspace*

11: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: $\mathrm{IFAIL} = 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters

may be useful even if IFAIL $\neq 0$ on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

If on exit IFAIL $= 1$ or 2, then G01JCF returns 0.0.

IFAIL $= 1$

On entry, N < 1,
or MAXIT < 1,
or C < 0.0.

IFAIL $= 2$

On entry, A has an element ≤ 0.0,
or MULT has an element < 1,
or RLAMDA has an element < 0.0.

IFAIL $= 3$

The central χ^2 calculation has failed to converge. This is an unlikely exit. A larger value of TOL should be tried.

IFAIL $= 4$

The solution has failed to converge within MAXIT iterations. A larger value of MAXIT or TOL should be used. The returned value should be a reasonable approximation to the correct value.

IFAIL $= 5$

The solution appears to be too close to 0 or 1 for accurate calculation. The value returned is 0 or 1 as appropriate.

7 Accuracy

The series (2) is summed until a bound on the truncation error is less than TOL. See Farebrother (1984) for further discussion.

8 Further Comments

None.

9 Example

The number of χ^2 variables is read along with their coefficients, degrees of freedom and non-centrality parameters. The lower tail probability is then computed and printed.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G01JCF Example Program Text
*     Mark 14 Release.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NMAX
      PARAMETER        (NMAX=4)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      real             C, P, PDF, TOL
      INTEGER          I, IFAIL, MAXIT, N
*     .. Local Arrays ..
      real             A(NMAX), RLAMDA(NMAX), WRK(1200)
      INTEGER          MULT(NMAX)
*     .. External Subroutines ..
      EXTERNAL         G01JCF
*     .. Data statements ..
      DATA             MAXIT, TOL/500, 0.0001e0/
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G01JCF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      WRITE (NOUT,*)
      WRITE (NOUT,*) '        A     MULT  RLAMDA'
   20 READ (NIN,*,END=60) N, C
      WRITE (NOUT,*)
      READ (NIN,*) (A(I),I=1,N)
      READ (NIN,*) (MULT(I),I=1,N)
      READ (NIN,*) (RLAMDA(I),I=1,N)
      IFAIL = -1
*
      CALL G01JCF(A,MULT,RLAMDA,N,C,P,PDF,TOL,MAXIT,WRK,IFAIL)
*
      IF (IFAIL.EQ.0 .OR. IFAIL.GE.4) THEN
         DO 40 I = 1, N
            WRITE (NOUT,99999) A(I), MULT(I), RLAMDA(I)
   40    CONTINUE
         WRITE (NOUT,99998) 'C = ', C, '    PROB = ', P
         GO TO 20
      END IF
   60 STOP
*
99999 FORMAT (1X,F10.2,I6,F9.2)
99998 FORMAT (1X,A,F6.2,A,F6.4)
      END
```

9.2 Program Data

```
G01JCF Example Program Data
 3    20.0                      :N  C
 6.0   3.0  1.0                 :A(I), I=1,N
 1     1    1                   :MULT(I), I=1,N
 0.0   0.0  0.0                 :RLAMDA(I), I=1,N
 2    10.0                      :N  C
 7.0   3.0                      :A(I), I=1,N
 1     1                        :MULT(I), I=1,N
 6.0   2.0                      :RLAMDA(I), I=1,N
```

9.3 Program Results

```
 G01JCF Example Program Results

         A     MULT  RLAMDA

       6.00     1     0.00
       3.00     1     0.00
       1.00     1     0.00
 C =  20.00    PROB = 0.8760

       7.00     1     6.00
       3.00     1     2.00
 C =  10.00    PROB = 0.0451
```